Environment Canada	Environnement Canada

National Hydrology Research Institute

NHRI PAPER NO. 17
IWD TECHNICAL BULLETIN NO. 125

A One-Dimensional Model of Unconfined Ground Water Flow over a Sloping Impermeable Base

A. Vandenberg

NHRI

NATIONAL HYDROLOGY RESEARCH INSTITUTE
INLAND WATERS DIRECTORATE
OTTAWA, CANADA, 1982

Environment Canada Environnement Canada

National Hydrology Research Institute

NHRI PAPER NO. 17

IWD TECHNICAL BULLETIN NO. 125

A One-Dimensional Model of Unconfined Ground Water Flow over a Sloping Impermeable Base

A. Vandenberg

NHRI

NATIONAL HYDROLOGY RESEARCH INSTITUTE
INLAND WATERS DIRECTORATE
OTTAWA, CANADA, 1982

Cat. No. En 36-503/125E

ISBN 0-662-12117-1

Contents

Page

ABSTRACT v

RÉSUMÉ v

INTRODUCTION 1

THE MODEL. 1

(a) Total flux through the boundary face is known 10

(b) Potential at the boundary is known at all time. . . . 11

FORTRAN CODING 11

Subroutines. 14

Core requirements. 14

INPUT DATA 14

EXAMPLES AND COMPARISON WITH ANALYTICAL SOLUTIONS. 17

(a) The linearized differential equation of ground water flow 17

Example 1: linear flow with constant recharge through the central compartment.. 18

Example 2 22

(b) The steady state differential equation. 23

Example 3 23

Example 4: steady state water table above a uniformly sloping base. 25

Example 5: a time dependent solution 27

REFERENCES 30

Tables

Page

Table 1. FORTRAN coding 12

Table 2. Comparison of analytical and finite-difference solutions, examples 1 and 2. 21

Table 3. Comparison of analytical and finite-difference steady state, example 3. 24

Table 4. Steady state flow over a sloping base, analytic and model solutions 27

Illustrations

Figure 1. Model of a linear unconfined aquifer. 2

Figure 2. Time dependent solution with uniformly rising head at x = 0. 29

Abstract

A mathematical one-dimensional model of unconfined ground water flow over a sloping impermeable base is described. The model employs a Crank-Nicholson implicit finite-difference formulation, in which the non-linearity of the differential equation owing to the varying saturated thickness is handled by iterative adjustment during each time step. Five examples of possible applications are described, and the model results are compared with analytical solutions.

Résumé

Un modèle mathématique unidimensionnel d'un écoulement souterrain non captif sur une base imperméable en pente est décrit. Ce modèle utilise une formule de différence finie implicite de Crank-Nicholson dans laquelle la non-linéarité de l'équation différentielle due à la variation de l'épaisseur saturée, est traitée par des ajustements itératifs au cours de chaque étape temporelle. Cinq exemples d'applications possibles sont décrits et les résultats fournis par le modèle sont comparés avec certaines solutions analytiques.

A One-Dimensional Model of Unconfined Ground Water Flow over a Sloping Impermeable Base

A. Vandenberg

INTRODUCTION

A study of the movement of leachate from a sanitary landfill near Gander, Newfoundland, led to the development of a model of ground water flow for the prediction of flow velocities on the basis of recorded rainfall data and measured aquifer characteristics. The landfill is situated a few miles from the Atlantic Coast in a terrain where metamorphic bedrock with extremely poor permeability is covered by approximately 10 m of unconsolidated sand and gravel. The bedrock surface slopes gently and uniformly toward the coast. Contour lines of the undisturbed water table are virtually parallel to each other and to the coast; thus ground water motion can adequately be described by a one-dimensional model. The model employs a Crank-Nicholson implicit finite-difference formulation; the non-linear differential equation for unconfined flow is first linearized by the assumption of constant saturated thickness, whereafter the saturated thickness is adjusted each time step by an iterative scheme.

THE MODEL

The most important simplification used in the derivation of the conceptual model (Fig. 1) is the assumption that, given the small angle of the slope of the impermeable base, the flow is essentially horizontal and in one direction only.



Figure 1. Model of a linear unconfined aquifer.

The strip of aquifer between two sub-parallel flow lines is subdivided into a number, n, of small compartments which are assigned a node W_i midway between each of the compartment boundaries. The compartments can be chosen such that the nodes coincide as much as possible with test or observation wells so that the test results can be taken as representative values over the compartment. However, to obtain a reasonable degree of accuracy, there usually will be more compartments than there are observation points.

To each of the compartments or nodes, the following parameters are assigned in which the subscript denotes the ith compartment (i=1, 2,...n):

L_i = distance between W_i and W_{i+1} (length),

A_i = width between bounding flow lines, measured at W_i (length),

D_i = height of the impermeable base above datum (length),

K_i = permeability (length/time),

S_i = specific storage (dimensionless),

R_i = rate of recharge from precipitation (positive) or evapotranspiration (negative) (length/time), and

h_i = saturated thickness (length).

The simulation procedure used in the model is basically a repeated application of the law of conservation of mass, stated as a water balance over each of the compartments. Starting with a known water level, and therefore a known volume stored in each compartment, the change in storage over a small period of time is calculated for each compartment:

$$\text{Increase in storage} = \text{algebraic sum of all inflows} \quad (1)$$

Inflows comprise infiltration over the area of the compartment and the ground water flow through the vertical cross sections separating the compartments. Infiltration, which may be negative, is calculated as the product of a given infiltration rate (length/time) and the area of the compartment. Ground water inflow or outflow is calculated by means of Darcy's law:

$$q = -KA'S' \quad (2)$$

where q = inflow rate (volume/time),

A' = cross-sectional area of surface through which flow takes place, perpendicular to the direction of flow, and

S' = the gradient of piezometric head or water table elevation (dimensionless).

The cross-sectional area, A', is the product of the width, A, of the compartment, and the saturated thickness, h:

$$A' = h\ A \quad (3)$$

Thus Equation 2 becomes

$$q = -\ K\ h\ AS' \quad (4)$$

Furthermore, K, h and A in Equation 4 refer to the values of these variables at the boundaries between the compartments, whereas in the model they are defined at the nodes themselves. Replacement of these variables by their mean values in Equation 4 is therefore a logical approximation, and the inflow, q_1, from the (i-1)th compartment to the ith compartment then becomes

$$q_1 = -(K_{i-1} + K_i)\ (h_{i-1} + h_i)\ (A_{i-1} + A_i)\ S'/8 \quad (5)$$

The inflow q_2, from the (i+1)th compartment to the ith compartment, is:

$$q_2 = +(K_{i+1} + K_i)(h_{i+1} + h_i)(A_{i+1} + A_i) S'/8 \qquad (6)$$

The gradient S', for the boundary between the (i-1)th compartment and the ith compartment, is approximated by:

$$[h_i + D_i - (h_{i-1} + D_{i-1})]/L_{i-1} \qquad (7)$$

and for the boundary between the ith and the (i+1)th compartment by:

$$[h_{i+1} + D_{i+1} - (h_i + D_i)]/L_i \qquad (8)$$

since the water level is the sum of the elevation of the base of the aquifer, D, and the saturated thickness.

In the terms of the water balance Equation 1 for infiltration and change in storage, the area of the ith compartment is needed; it is approximated as a trapezoid with height

$$(L_i + L_{i-1})/2$$

and parallel sides of lengths

$$(A_i + A_{i-1})/2$$

and

$$(A_i + A_{i+1})/2$$

respectively. Thus the area becomes

$$\text{Area}_i = (A_{i-1} + 2A_i + A_{i+1})\ (L_i + L_{i-1})/8 \tag{9}$$

Thus the total recharge to the ith compartment per unit of time is

$$\text{Recharge} = R_i\ (A_{i-1} + 2A_i + A_{i+1})\ (L_i + L_{i-1})/8 \tag{10}$$

Similarly, the increase in storage inside the compartment due to a rising or falling water table, per unit of time, is

$$\Delta\ \text{Storage} = S_i\ (A_{i-1} + 2A_i + A_{i+1})(L_i + L_{i-1})(H_i^* - h_i)/(8\Delta t) \tag{11}$$

where H_i^* = the saturated thickness at the end of a time interval Δt.

The complete water balance of the ith compartment can now be written:

$$q_1 + q_2 + \text{recharge} = \text{increase in storage}$$

or

$$(h_{i-1} + h_i)(A_{i-1} + A_i)(K_{i-1} + K_i)(h_{i-1} - h_i + D_{i-1} - D_i)/L_{i-1}$$

$$+ (h_i + h_{i+1})(A_i + A_{i+1})(K_i + K_{i+1})(h_{i+1} - h_i + D_{i+1} - D_i)/L_i \tag{12}$$

$$+ R_i(A_{i-1} + 2A_i + A_{i+1})(L_i + L_{i-1}) = (S_i/\Delta t)(A_{i-1} + 2A_i + A_{i+1})(L_i + L_{i-1})(H_i^* - h_i)$$

Using the formulation of (12), H_i^* could be calculated explicitly for each of the nodes and for successive time increments. For very small time steps such a procedure would be acceptable, but necessarily slow; for larger time steps the formulation of (12) must be modified, since (12) implies that throughout the time step, Δt, the saturated thickness, from which both the cross-sectional area of the flow and the potential gradient are determined, may be assumed to be equal to the saturated thickness at t_0, the beginning of the time step. Instead of this coarse approximation for the true but unkown value of h(t), it is more accurately replaced by the linear expression:

$$h(t) = h + (H^* - h)(t - t_0)/\Delta t \tag{13}$$

The average value of h(t) over Δt is therefore

$$h(t) = (1/\Delta t)\int_{t_0}^{t_0+\Delta t} [h + (H^* - h)(t - t_0)/\Delta t]dt = (H^* + h)/2 \tag{14}$$

Substitution of (14) in (12) renders the equation implicit, i.e., unknown values H_i^*, H_{i-1}^* and H_{i+1}^* appear in the equation for the ith element. Furthermore, since H^* appears as part of the hydraulic head, $h_i + D_i$, as well as a factor of the cross-sectional area, the equation is no longer linear, and direct solution becomes very awkward. In the present model the non-linearity is handled as follows:

(a) A distinction is made between (1) the saturated thickness appearing in the expression for the cross-sectional area, for which the designation h is retained, and (2) the saturated thickness appearing in the expression for hydraulic head, for which the notation H will be used.

(b) Initially, h will be assumed constant with a value equal to the value of h at the beginning of the time step, whereas H will be averaged over the time step, i.e.,

$$H_{ave} = (H + H^*)/2$$

where H now represents the value at the beginning of the time step and is therefore known, whereas H* represents the value of H at the end of the time step.

(c) The set of linear equations resulting from (1) for all nodes is solved for H*.

(d) The value of h is now replaced by (H + H*)/2 and the set of linear equations is again solved, resulting in improved values of H*.

(e) Step (d) is continued until the new value of H* differs from the last calculated value by less than a predetermined amount.

(f) The new values H* replace the old values of H and the process is repeated for the next time step.

Replacement of h by (H* + H)/2 in the term for hydraulic gradient of (1) results in:

$$(h_{i-1}+h_i)(K_{i-1}+K_i)(A_{i-1}+A_i)\,[H_{i-1}{}^*-H_i{}^*+H_{i-1}-H_i+2(D_{i-1}-D_i)]\,/(2L_{i-1})$$
$$+\,(h_i+h_{i+1})(K_i+K_{i+1})(A_i+A_{i+1})\,[H_{i+1}{}^*-H_i{}^*+H_{i+1}-H_i+2(D_{i+1}-D_i)]\,/(2L_i) \quad (15)$$
$$+\,R_i(A_{i-1}+2A_i+A_{i+1})(L_i+L_{i-1}) = (S_i/\Delta t)(A_{i-1}+2A_i+A_{i+1})(L_i+L_{i-1})(H_i{}^*-H_i)$$

Rearranging (15) with the H* on the left-hand side and the known terms on the right-hand side gives

$$-\,U_i\;H_{i-1}{}^* + H_i{}^*\;(U_i + S_iP_i/\Delta t + U_{i+1}) - U_{i+1}\;H_{i+1}{}^*$$
$$= U_i\;[H_{i-1} - H_i + 2\;(D_{i-1} - D_i)] + U_{i+1}\;[H_{i+1} - H_i + 2\;(D_{i+1} - D_i)] \quad (16)$$
$$+\,P_i\;(R_i + S_iH_i/\Delta t)$$

where

$$U_i = (h_{i-1} + h_i)(K_{i-1} + K_i)(A_{i-1} + A_i)/(2L_{i-1}) \quad (17)$$

and

$$P_i = (A_{i-1} + 2A_i + A_{i+1})(L_{i-1} + L_i) \quad (18)$$

Equation 16 is valid for any of the interior elements. For the end elements a slightly different equation is required. Depending on which assumption is made, it may take either of two forms: either (a) the total flux through the aquifer boundary face is known at all time or (b) the potential at the aquifer boundary is known at all time. Either of these conditions must be fulfilled for the problem to have a unique solution.

(a) Total Flux through the Boundary Face Is Known

Introducing the notation:

Q_o = total influx through the aquifer boundary adjacent to node W_1,

Q_m = total influx through the aquifer boundary adjacent to node W_n,

L_o = distance between W_1 and the adjacent boundary, and

L_m = distance between W_n and the adjacent boundary.

The term Q_o replaces the term q_1 (Equation 5) in the water balance over the first element, giving

$$H_1^* [U_2 + 8S_1A_1 (L_o + L_1/2)/\Delta t] - U_2H_2^*$$
$$= 8Q_o + U_2 [H_2 - H_1 + 2(D_2 - D_1)] + 8A_1(R_1 + S_1H_1/\Delta t)(L_o + L_1/2) \quad (19)$$

The term Q_m replaces the term q_2 (Equation 6) in the water balance over the last element, giving

$$- H_{m-1}^* U_m + H_m^* [U_m + (8S_mA_m/\Delta t)(L_m + L_{m-1}/2)]$$
$$= 8Q_m + U_m [H_{m-1} - H_m + 2 (D_{m-1} - D_m)] + 8A_m(R_m + S_mH_m/\Delta t)(L_m + L_{m-1}/2) \quad (20)$$

(b) Potential at the Boundary Is Known at All Time

Given the saturated thickness at the aquifer boundary adjacent to node W_1 , H_o, the equation for the water balance of the first element can be stated:

$$H_1{*}\,[4A_1K_1H_o/L_o + U_2 + 8A_1S_1(L_o + L_1/2)/\Delta t] - U_2H_2{*}$$
$$= 4A_1K_1H_o(2H_o-H_1)/L_o+U_2\,[H_2-H_1+2(D_2-D_1)] + 8A_1(L_o+L_1/2)(R_1+S_1H_1/\Delta t) \qquad (21)$$

and if H_L is the saturated thickness at the aquifer boundary adjacent to the last node, the equation for the last element becomes:

$$- U_mH_{m-1}{*} + H_m{*}\,[4A_mK_mH_L/L_m + U_m + 8A_mS_m(L_m + L_{m-1}/2)/\Delta t]$$
$$= 4A_mK_mH_L(2H_L - H_m)/L_m + U_m\,[H_{m-1} - H_m + 2(D_{m-1} - D_m)] \qquad (22)$$
$$+ 8A_m(L_m + L_{m-1}/2)(R_m + S_mH_m/\Delta t)$$

FORTRAN CODING

FORTRAN IV coding for a computer program used on the Control Data Cyber 74 computer at the Computer Science Centre, Department of Energy, Mines and Resources, Ottawa, is given in Table 1.

Table 1. FORTRAN Coding

```
      PROGRAM ONED        73/74    OPT=1                        FTN 4.8+528         03/11/81  14.37.39

            PROGRAM ONED(INPUT,OUTPUT)
      C*************************************************************************
      C***
      C***  PROGRAM ONED SIMULATES ONE-DIMENSIONAL UNCONFINED GROUNDWATER FLOW
      C***  OVER AN IMPERMEABLE,SLOPING BASE.THE SOLUTION TECHNIQUE IS A C-N
      C***  FINITE DIFFERENCE METHOD WITH ITERATIVE ADJUSTMENT OF THE
      C***  SATURATED THICKNESS.CONSTANT HEAD OR CONSTANT FLUX BNDRY-CONDITIONS
      C***  MAY BE SELECTED INDEPENDENTLY AT EACH BNDRY.
      C***
      C*************************************************************************
      C***
            DIMENSION AL(50),A(50),AK(50),H(50),HA(50),HD(50),D(50),S(50),
           1U(50),P(50),R(50),B(200),RHS(50)
      C********         LIST OF INPUT VARIABLES
      C***  DATA INPUT IS LIST-DIRECTED AND FREE FORMAT,BUT FOR CLARITY WILL
      C***  BE DESCRIBED HERE AS APPEARING ON 3 DISTINCT CARDS   *************
      C*************************************************************************
      C***                     CARD 1
      C*** N      =NUMBER OF NODES
      C*** MAX    =MAXIMUM NUMBER OF ITERATIONS IN EACH TIMESTEP
      C*** TEST   =ACCURACY LIMIT OF ITERATIONS
      C*** ALO    =DISTANCE FROM FIRST NODE TO NEAREST BOUNDARY
      C*** ALL    =DISTANCE FROM LAST  NODE TO NEAREST BOUNDARY
      C*** IS     = IS=0,THEN ON BOTH BOUNDARIES FLUX IS GIVEN
      C***            =1,THEN FLUX GIVEN ON BNDRY NEAR NODE 1,HEAD ON BNDRY NEAR NODE N
      C***            =2,THEN HEAD GIVEN ON BNDRY NEAR NODE 1,FLUX ON BNDRY NEAR NODE N
      C***            =3,THEN ON BOTH BNDRIES HEAD IS PRESCRIBED
      C*** ID     =SKIPFACTOR FOR PRINTOUT OF WATERLEVELS AND SAT.THICKNESS
      C***                     CARD 2
      C*** A(I),I=1,N  =COMPARTMENT WIDTH
      C*** AK(I),I=U,N =PERMEABILITY
      C*** H(I),I=1,N  =SATURATED THICKNESS AT TIME ZERO
      C*** D(I),I=1,N  =ELEVATION OF IMPERMEABLE BASE
      C*** S(I),I=1,N  =STORATIVITY
      C*** AL(I),I=1,N-1 =DISTANCE BETWEEN NODES
      C***                     CARD 3
      C*** DT   =INITIAL LENGTH OF TIMESTEP
      C*** TINC =MULTIPLICATION FACTOR APPLIED TO DT AT EACH TIMESTEP
      C*** TMAX =MAXIMUM TIME OF SIMULATION WITH THIS SET  OF DATA
      C*** Q1   =FLUX OR HEAD AT BNDRY NEAR NODE 1 , DEPENDING ON VALUE OF IS
      C*** Q2   =FLUX OR HEAD AT BNDRY NEAR NODE N , DEPENDING ON VALUE OF IS
      C*** R(I),I=1,N.  INFILTRATION RATE
      C***
      C*** REPEAT CARD 3 AS OFTEN AS YOU WISH,WITH INCREASING TMAX
      C*** LAST CARD MUST CONTAIN (N+5) VALUES OF WHICH THE FIRST MUST BE ZERO.
      C***
            READ*,N,MAX,TEST,ALO,ALL,IS,ID
            PRINT 130
      130   FORMAT(*1SOLUTION OF ONE-DIMENSIONAL GROUNDWATER FLOW*//)
            PRINT 131,N,MAX,TEST,ALO,ALL,IS,ID
        131 FORMAT(* NUMBER OF NODES                =*,I3/
           1         * MAXIMUM NUMBER OF ITERATIONS =*,I3/
           2         * ACCURACY                     =*,E12.5/
           3         * DISTANCE TO LHS BOUNDARY     =*,E12.5/
           4         * DISTANCE TO RHS BOUNDARY     =*,E12.5/
           5         * BOUNDARY CONDITIONS,SET      =*,I2 /
           6         * SKIP FACTOR                  =*,I2//)
            N1=N-1 $ NN=N1*3 $ TM=0
            READ*,(A(I),I=1,N),(AK(I),I=1,N),(H(I),I=1,N),(D(I),I=1,N),
           1(S(I),I=1,N), (AL(I),I=1,N1) $DO 22 I=2,N1
      22    P(I)=(A(I-1)+2.*A(I)+A(I+1))*(AL(I-1)+AL(I))
            PRINT 300 $ DO 369 I=1,N
        369 PRINT 400,I,A(I),AK(I),           H(I),D(I),S(I),AL(I)
        400 FORMAT(1X,I5,6E12.4)
        300 FORMAT(4X,*I*,5X,*A=WIDTH*,5X,*AK=PERM. H=SAT.THICK D=BASE ELEV S=
           1STOR. COEF AL=LENGTH*)
      3     READ*,DT,TINC,TMAX,Q1,Q2,(R(I),I=1,N)$IF(DT.EQ.0.)STOP
            PRINT 173,DT,TINC,TMAX        ,Q1,Q2
      173   FORMAT(//* INITIAL T-STEP,TIMESTEP-INCREMENT,MAXIMUM TIME,Q1 AND Q
```

```
          12*//          ,5E15.4/)
           PRINT 174,(R(I),I=1,N)
       174 FORMAT(1X,10E12.4)
      C***
      C*** CALCULATE COEFFICIENTS OF MATRIX-EQUATION
      C***
      9    ICNT=0 $ DO 1 I=1,N $ HD(I)=H(I)
      1    HA(I)=H(I)
      10   DO 2 I=2,N $ U(I)=(HA(I-1)+HA(I))*(AK(I-1)+AK(I))*
          1(A(I-1)+A(I))/(2.*AL(I-1))
      2    CONTINUE $ IF(IS.GT.1)GOTO 12
           B(1)=U(2)+8.*S(1)*A(1)*(ALO+AL(1)/2.)/DT $ B(2)=-U(2)
           RHS(1)=8.*Q1+U(2)*(H(2)-H(1)-2.*(D(1)-D(2)))+
          1 8.*(R(1)+S(1)*H(1)/DT)*A(1)*(ALO+AL(1)/2.)
           IF(IS.EQ.1)GOTO 14
        15 B(NN)=-U(N)
           B(NN+1)=U(N)+8.*S(N)/DT*A(N)*(ALL+AL(N1)/2.)
           RHS(N)=8.*Q2+U(N)*(H(N1)-H(N)+2.*(D(N1)-D(N)))
          1 +8.*A(N)*(AL(N1)/2.+ALL)*(R(N)+S(N)*H(N)/DT)
        13 DO 4 I=2,N1 $ II=(I-1)*3 $ B(II)=-U(I) $ B(II+2)=-U(I+1)
           B(II+1)=U(I)+U(I+1)+S(I)*P(I)/DT
      4    RHS(I)=U(I)*(H(I-1)-H(I)+2.*(D(I-1)-D(I)))
          1+U(I+1)*(H(I+1)-H(I)+2.*(D(I+1)-D(I)))
          2+P(I)*(S(I)*H(I)/DT+R(I))
      C****
      C***    SOLVE MATRIX-EQUATION
      C***
           CALL MAS008(RHS,B,N,1,1,1,1.E-4,IER)
           IF(ICNT.GT.MAX)GOTO 7 $ DO 6 I=1,N1
           IF(RHS(I).LT.0.)RHS(I)=.0001
           IF(ABS(RHS(I)-HD(I)).GT.TEST)GOTO 11
      6    CONTINUE
      7    DO 8 I=1,N
           B(I)=RHS(I)+D(I)
      8    H(I)=RHS(I) $ TM=TM+DT $DT=DT*TINC
           IF((TM+DT).GT.TMAX)DT=TMAX-TM
           PRINT 100,TM,ICNT,IER
           PRINT 600
           PRINT 200,(H(I),I=1,N,ID)
           PRINT 500
           PRINT 200,(B(I),I=1,N,ID)
       500 FORMAT(/*  WATERLEVELS*/)
       600 FORMAT(/*  SATURATED THICKNESS*/)
           IF(TM.GE.TMAX)GOTO3 $ GOTO 9
       100 FORMAT(/* TIME*          ,E12.4,2I5/)
      200  FORMAT(1X,10E11.4)
      11   DO5 I=1,N $ HA(I)=(H(I)+RHS(I))/2.
      5    HD(I)=RHS(I) $ ICNT=ICNT+1 $ GOTO 10
      12   FA=4.*A(1)*AK(1)*Q1/ALO $ FB=8.*A(1)*(ALO+AL(1)/2.)
           B(1)=FA +U(2)+FB*S(1)/DT $B(2)=-U(2)
           RHS(1)=FA*(2.*Q1-H(1))+U(2)*(H(2)-H(1)+2.*(D(2)-D(1)))
          1+FB*(R(1)+S(1)*H(1)/DT)
           IF(IS.EQ.2)GOTO 15
        14 B(NN)=-U(N) $ FA=4.*A(N)*AK(N)*Q2/ALL
           FB=8.*A(N)*(ALL+AL(N-1)/2.) $ B(NN+1)=FA+U(N)+FB*S(N)/DT
           RHS(N)=FA*(2.*Q2-H(N))+U(N)*(H(N-1)-H(N)+2.*(D(N-1)-D(N)))
          1+FB*(R(N)+S(N)*H(N)/DT) $ GOTO 13 $ END
```

Subroutines

The main program ONED - for one-dimensional - calls the subroutine MAS008 for the solution of a system of linear equations with band structure; MAS008 is identical with subroutine GELB of the IBM Scientific Subroutine Package (IBM, 1966).

Core Requirements

The program requires 40 K words of memory.

INPUT DATA

Input data are in free format, and data are given in the following order. Even though card boundaries are disregarded in free format, for convenience in the description each READ - statement is represented as one card.

Card 1

Item 1:	N =	the number of compartments (integer)
Item 2:	MAX =	maximum number of iterations at each time step (integer)
Item 3:	TEST=	accuracy criterion for the iterations: If the difference between the values of H calculated during the last and second last iterations is

less than TEST for all nodes, the iteration is terminated, results for the time step are printed, and calculation for the next time step starts.

Item 4: ALO = distance between the first node and the adjacent aquifer boundary (length)

Item 5: ALL = distance between the last node and the adjacent aquifer boundary (length)

Item 6: IS = code number indicating the boundary conditions:

IS = 0: On both boundaries the flux is known.

IS = 1: Flux specified on boundary near node 1, head is known on boundary near node N.

IS = 2: Head specified on boundary near node 1, flux specified on boundary near node N.

IS = 3: Head specified on both boundaries.

Item 7: ID = skip factor for printout of water table elevations at selected nodes. If ID = 1, values for all nodes will be printed; if ID > 1, values will be printed only for nodes 1, 1+ID, 1+2ID, etc.

Card 2

A_i, i = 1,N: compartment widths (Length)

AK_i, i = 1,N: permeabilities (Length/time)

H_i, i = 1,N: saturated thickness (Length)

D_i, i = 1,N: elevation of the impermeable base (Length)

S_i, i = 1,N: storativities (Dimensionless)

AL_i, i = 1,N-1: distances between nodes, only N-1 values (Length)

Card 3

DT = initial length of time step (Time)

TINC = multiplication factor applied to DT at each time step in order to increase the length of the time step gradually (Dimensionless)

TMAX = maximum time of simulation with this set of data (Time)

Q_1 = if IS = 0 or IS = 1, the constant flux through the aquifer boundary into the first element ($Length^3$/Time)

if IS = 2 or IS = 3, the constant saturated thickness at the aquifer boundary near the first element (Length)

Q_2 = if IS = 0 or IS = 2, the constant flux into the last element ($Length^3$/Time)

if IS = 1 or IS = 3, the constant saturated thickness at the boundary near the last element (Length)

R_i, i = 1,N the infiltration into each element from the top (Length/Time)

Card 3 can be repeated a number of times, each time with different values of R_i, Q_1, Q_2, as well as the time step parameters. Note that each TMAX must be greater than on the previous card.

Last Card: To stop the program, the value of DT on the last card must be zero. Note that even if DT = 0, the program still expects values for the other variables on this card. This can be achieved simply by using the repeated input facility available under free format input; thus the last card can contain simply: 100 * 0.

EXAMPLES AND COMPARISON WITH ANALYTICAL SOLUTIONS

(a) The Linearized Differential Equation of Ground Water Flow

If the change in saturated thickness with time and place is small in comparison with the total saturated thickness, H, the latter may be assumed to be a constant. The differential equation for horizontal unconfined ground water flow then becomes identical with the equation for confined flow. In the one-dimensional case we have:

(1) The non-linear formulation

$$(\partial/\partial x)(Kh\partial h/\partial x) = S\ \partial h/\partial t \qquad (23)$$

(2) The linear formulation

$$HK\partial^2 h/\partial x^2 = S\ \partial h/\partial t \qquad (24)$$

Example 1: Linear Flow with Constant Recharge through the Central Compartment

The first example is a simulation of a horizontal aquifer (D = constant) with constant permeability. The initial head in the aquifer is constant, and the boundaries are assumed to remain at a constant head equal to the initial head in the aquifer. The ground water is thus in static equilibrium at the beginning of the simulation, when a constant recharge q is applied to the central compartment. Furthermore, the parameters L, A, K and S are uniform throughout the aquifer.

This simulation is equivalent to the differential Equation 24 - provided the change in saturated thickness remains small compared with the total saturated thickness - with the boundary conditions:

$$h(x, 0) = h(s, t) = h_0 \qquad (25)$$

$$AHK\ (\partial h/\partial x) = -qAL/2 \qquad \text{at } x = 0$$

or

$$\partial h/\partial x = qL/(2HK) \qquad \text{at } x = 0 \qquad (26)$$

and x = s is the location of either boundary of the aquifer. On account of the symmetry of the problem the recharge, q, is divided evenly between the left- and the right-hand section of the aquifer. For points near the central compartment, far away from either boundary, and for a short period after the recharge commences, the boundary may furthermore be assumed to be at infinity. Under these assumptions, the solution to the boundary value problem has been found to be (for example, Glover, 1978).

$$h = h_0 + [qLx/(2DK\sqrt{\pi})]\ G(u)$$

where $u = (x^2 S/4KHt)^{1/2}$

x = distance to the recharge point, and

$$G(u) = \int_u^{\infty} \exp(-y^2)/y^2 dy \qquad (27)$$

For the simulation of flow with recharge through the central node the following parameters were used:

N	=	number of nodes	= 49
MAX	=	maximum number of iterations	= 100
TEST	=	accuracy of iterative adjustment of saturated thickness	= 0.0001 m
ALO	=	distance to l.h.s. boundary	= 10 m
ALL	=	distance to r.h.s. boundary	= 10 m
IS	=	code for boundary conditions (i.e. head specified on both boundaries)	= 3
ID	=	skip factor for printing of results (i.e. results at all nodes are to be printed)	= 1

A_i, $i = 1, 49 =$ 100 m

K_i, $i = 1, 49 =$ 1 m/day

h_i, $i = 1, 49 =$ 10 m

D_i, $i = 1, 49 =$ 0

S_i, $i = 1, 49 =$ 0.15

L_i, $i = 1, 48 =$ 10 m

DT =	initial time step	= 1 day
TINC =	time step increment	= 1.03
TMAX =	simulated period	= 500 days
Q_1 =	constant head, l.h.s. boundary	= 10 m
Q_2 =	constant head, r.h.s. boundary	= 10 m

Recharge: R_i, $i = 1,2...24 = 0$

$R_{25} = 0.015$ m/day

$R_i, i = 26,27...49 = 0$

The input for this simulation is as follows:

card 1 = 49 100 0.0001 10 10 3 1

card 2 = 49*100 49*1 49*10 49*0 49*.15 48*10

card 3 = 1 1.03 500 10 10 24*0 0.015 24*0

card 4 = 100*0

In Table 2 the results for node No. 21, 40 m from the recharged compartment, are compared with the analytical solution:

$$h - h_o = + \frac{(0.015)(10)(40)}{(2)(10)(1)\sqrt{\pi}} G(2.45/t^{1/2})$$

or

$$h - h_o = 0.1693\ G(2.45/t^{1/2}) \tag{28}$$

Table 2. Comparison of Analytical and Finite-Difference Solutions, Examples 1 and 2

t (days)	u	G(u)*	$h-h_o$ (m) Analytic solution	$h-h_o$ (m) Finite diff. example 1	$h-h_o$ (m) Finite diff. example 2
1	2.45	7×10^{-5}	0	0	0
2.03	1.72	0.0036	0.0006	0	0
3.09	1.39	0.0165	0.003	0	0
4.18	1.20	0.0388	0.007	0.01	0.01
5.31	1.06	0.0686	0.012	0.01	0.01
6.47	0.963	0.104	0.017	0.02	0.02
8.89	0.822	0.185	0.031	0.03	0.03
11.5	0.724	0.276	0.047	0.05	0.05
15.6	0.620	0.424	0.072	0.07	0.07
20.2	0.546	0.580	0.098	0.10	0.10
28.7	0.458	0.856	0.145	0.14	0.14
34.4	0.418	1.03	0.174	0.17	0.17
50	0.347	1.46	0.247	0.25	0.25
69.2	0.295	1.93	0.327	0.32	0.32
100	0.245	2.56	0.433	0.43	0.43
152	0.199	3.45	0.584	0.58	0.58
201	0.173	4.18	0.708	0.70	0.70
301	0.141	5.46	0.924	0.89	0.91
403	0.122	6.54	1.107	1.04	1.08
500	0.110	7.42	1.256	1.14	1.24

*From Glover (1978)

Note that toward the end of the simulation, the finite-difference results are less than the corresponding analytical results due to the effect of the finite distance to the boundary.

Example 2

To test the numerical solution when flux rather than constant head is specified on the boundaries, Equation 24 and condition (25) can be used, but condition (26) has to be slightly modified. Since Q_1 represents the total influx through the vertical face at $x = 0$, we have

$$Q_1 = AHK\,\partial h/\partial x$$

or

$$\partial h/\partial x = Q_1/AHK$$

Then the analytic solution becomes

$$h - h_0 = Q_1 x/(AHK\sqrt{\pi})\ G(u)$$

By assigning the value of 7.5 m^3/day to Q_1, the numerical values for identical values of t will be identical with those given for the analytical solution in example 1 (Equation 28). The results for the 4th node, 40 m from the recharge boundary, are given in the last column of Table 2. Note that since the constant head boundary on the right-hand side is now approximately twice as far away from the recharge point, the deviation, for $t > 200$ min, of the finite-difference solution from the analytic solution is far less severe than in the first example.

(b) The Steady State Differential Equation

In ground water systems in which the flow is at equilibrium - no change in head with time - Equation 24 simplifies to

$$(\partial/\partial x)(Kh\ \partial h/\partial x) = 0 \tag{29}$$

and the non-linearity presents no problem. Equation 29 can be solved:

$$h^2 = ax + b \tag{30}$$

where a and b are constants depending on the boundary conditions; if $h(x = 0) = h_o$ and $h(x = L) = h_L$, then (30) becomes

$$h^2 = h_o^2 + (h_L^2 - h_o^2)x/L \tag{31}$$

Example 3

This example simulates how the head in a horizontal aquifer, originally in static equilibrium with a constant head of 5 m everywhere in the aquifer, will reach a new dynamic equilibrium when the head at one end of the aquifer is suddenly raised to 10 m and kept constant at 10 m thereafter; the necessary input cards are shown below:

```
19  100  0.0001  10  10  3  1
19*100  19*1  19*5  19*0  19*1  18*10
1  1.1  20000  5  10  19*0
100*0
```

A steady state is reached after 442 days; the head at each of the nodes calculated from (31) and those calculated with the finite-difference model are given in Table 3.

Table 3. Comparison of Analytical and Finite-Difference Steady State, Example 3

Node No.	x (m)	h (steady state) computed from Eq. 31 (m)	h (steady state, computed by model) (m)
1	10	5.362	5.374
2	20	5.701	5.712
3	30	6.021	6.031
4	40	6.325	6.334
5	50	6.614	6.623
6	60	6.892	6.900
7	70	7.159	7.166
8	80	7.416	7.422
9	90	7.665	7.671
10	100	7.906	7.911
11	110	8.139	8.144
12	120	8.367	8.371
13	130	8.588	8.592
14	140	8.803	8.807
15	150	9.014	9.017
16	160	9.220	9.222
17	170	9.421	9.423
18	180	9.618	9.620
19	190	9.811	9.813

Example 4: Steady State Water Table above a Uniformly Sloping Base

Given an aquifer underlain by a uniformly sloping base such that:

$D = \beta x$

where $\beta << 1$

the quasi-horizontal flow can be characterized by the differential equation

$$\partial/\partial x\,[h\,\partial(h + \beta x)/\partial x] = (S/K)(\partial h/\partial t)$$

and for the steady state

$$\partial/\partial x\,[h\,(\partial h/\partial x + \beta)] = 0$$

Then

$$h\,(\partial h/\partial x + \beta) = C_1 = -q'/K \qquad (32)$$

where q' = the constant flux per unit width at any given cross section ($length^2/time$).

Equation 32 can be rewritten as

$$dx = h/(C_1 - \beta h)\,dh$$

from which

$$\beta^2 x = \beta h + C_1 \ln (C_1 - \beta h) + C_2$$

If, for example, the saturated thickness at x = L is kept constant

$$h\ (x=L) = h_L$$

then

$$C_2 = \beta^2 L - \beta h_L - C_1 \ln (C_1 - \beta h_L)$$

and

$$x = L+(h-h_L)/\beta + (C_1/\beta^2) \ln [(C_1 - \beta h)/(C_1 - \beta h_L)] \tag{33}$$

For L=0, i.e., the saturated thickness specified at x=0

$$h(x=0) = h_o$$

$$x = (h-h_o)/\beta + (C_1/\beta^2) \ln [(C_1-\beta h)/(C_1-\beta h_o)] \tag{34}$$

Thus, with the aid of (33) or (34), the steady state saturated thickness at points in the aquifer can be calculated by inversion, i.e., arbitrary values of h are assumed and the corresponding value of x is calculated. Values of h at arbitrary values of x can then be found by interpolation. These calculations were carried out for the simulated aquifer, with the following characteristics and inputs:

K = 1 m/day

S = 0.1

A = 1 m

β = 0.1

L = 100 m

$Q_1 = 1\ m^3/day$, corresponding to $C_1 = -1$ m

$h_L = 5$ m

The model was run three times, with 9, 19 and 49 uniformly spaced nodes. The initial saturated thickness was 25 m at all nodes. Table 4 gives the steady state solutions obtained from (33) and from the three model runs, showing how the accuracy improves when the number of nodes is increased.

Table 4. Steady State Flow over a Sloping Base, Analytic and Model Solutions

x (m)	Saturated thickness, h (m)			
	Analytic solution	Model solutions: 9 nodes	Model solutions: 19 nodes	Model solutions: 49 nodes
0	22.834	-	-	-
10	21.37	21.01	21.18	21.29
20	19.88	19.51	19.67	19.80
30	18.36	17.98	18.15	18.28
40	16.78	16.40	16.57	16.70
50	15.15	14.76	14.94	15.07
60	13.47	13.04	13.23	13.37
70	11.66	11.21	11.41	11.57
80	9.74	9.23	9.45	9.62
90	7.60	7.00	7.26	7.45
100	5.00	-	-	-

Example 5: A Time Dependent Solution

Although in general Equation 23 cannot be solved analytically, a particular solution can be found on inspection; supposing $h(x, t)$ to be of the form

$$h = ax + bt \tag{35}$$

substitution in (23) gives

$$a = \pm (Sb/K)^{1/2}$$

The constant b is determined by the boundary condition at $x = 0$:

$$h(x = 0, t) = bt$$

that is, the head at $x = 0$ rises linearly with time from $h = 0$ at $t = 0$. Assuming the solution with negative a, we note further that positive values of the head are only possible for

$$x < t(Kb/S)^{1/2}$$

and for $x > t(Kb/S)^{1/2}$ the solution (35) must be replaced by the trivial solution

$$h = 0 \tag{36}$$

Thus:

For $x < t(Kb/S)^{1/2}$: $h = bt - x(Sb/K)^{1/2}$

For $x > t(Kb/S)^{1/2}$: $h = 0$

is the complete solution of the differential equation and the initial and boundary conditions:

$$h(x = 0, t) = bt$$
$$h(x = \infty, t) = 0$$
$$h(x, t = 0) = 0$$

This problem was simulated by making a slight adjustment to the present program, permitting the value of Q_1, representing the head at x = 0 under condition IS = 3, to be set equal to the time at the end of each time step multiplied by a constant factor, b. Figure 2 shows the simulated heads at t = 500, which fall midway between the analytic solution for t = 456, the beginning of the time step, and for t = 500. Thus, reducing the time step obviously will result in convergence to the true solution.



Figure 2. Time dependent solution with uniformly rising head at x = 0.

REFERENCES

Glover, R.E. 1978. Transient Ground Water Hydraulics. Water Resources Publications, P.O. Box 303, Fort Collins, Col., 80522, U.S.A.

IBM. 1966. Scientific Subroutine Package; Programmers Manual. IBM Technical Publications Dep., 112 East Post Road, White Plains, N.Y., 10601, U.S.A.